
Name pdemagcomp.f 
Contributor James Dewey, U. S. Geological Survey/NEIC, MS 966, Denver,  

CO 80225-0046; E-mail:  jdewey@usgs.gov 
Versions November 2011; DOI: 10.2312/GFZ.NMSOP-2_PD_3.1 

 
The following FORTRAN program, pdemagcomp.f, is based on a code for computing mb 
magnitude in the PDE bulletin of the USGS/NEIC. The original author is unknown. The 
program incorporates a procedure, mbmag, that has been used for decades to compute mb 
magnitude for the USGS/NEIC.  It is included to provide the source code for mbmag and to 
show how mbmag is used.  Program pdemagcomp.f was writtten by James W. Dewey, 
USGS/NEIC, for a limited research objective. 
 
pdemagcomp_f. 
 
c  program pdemagcomp.f 
 
c  calculates a Gutenberg-Richter mb, given epicentral distance, 
c  event focal depth, amplitude in nanometers and period in seconds 
 
c  requires 'qfile', the table of Gutenberg-Richter Q values 
 
      common q(108,17) 
      character*1 mbflg 
      real mb 
 
      open(unit = 23, status = 'old', file = 'qfile') 
      read (23,32,end = 33) ((q(i,j),j=1,17),i=1,108)  
   32 format(17(x,f4.2)) 
   33 continue 
 
      type *, 'enter event focal depth, f5.1' 
      read *, depth 
      type *, 'enter station epicentral distance, degrees, f6.2' 
      read *, delta 
      type *, 'enter amplitude in nanometers, f7.1' 
      read *, amp 
      type *, 'enter period in seconds, f3,1' 
      read *, t 
      type *, 'enter P-wave travel-time residual in seconds, f5.1' 
      read *, res 
      call mbmag(t,amp,mbflg,mb,depth,delta,res) 
      type *, 'mb = ', mb, 'mbflag = ', mbflg 
      stop 
      end 
 
      subroutine mbmag(t,amp,mbflg,mb,depth,delta,res) 
 
c     explanation of parameters (J. Dewey, 3/31/99) 
c     t = P-wave period 
c     amp = amplitude in nanometers 
c     mbflg = 'n' -- computed magnitude not reported for this station 
c           = 'R' or 'r' -- computed magnitude reported for this station, 
c                           but not used to compute average event magnitude 
c     mb = short-period P-wave magnitude 
c     depth = focal depth in km 
c     delta = epicentral distance in degrees 

c     q = magnitude calibration function for Gutenberg-Richter mb.  tabulated  
c         from 2 to 109 degrees, at depths of 0,25,50,75,100,150,...,650,700 km. 
c     res = travel-time residual, in seconds, of P-wave with which 
c           this magnitude is associated 
 
      common q(108,17) 
      character*1 mbflg 
      real mb 
c 
      mb=0. 
      if(t.lt..1.or.t.gt.9.) go to 11 
      if(amp.eq.0.) go to 11 
      if(delta.lt.5..or.delta.gt.109.) go to 11 
c     if(delta.lt.5..and.depth.ne.0.) go to 11 
      if(abs(res).gt.10..or.depth.gt.700.) go to 11 
c 
      do 7 k=1,17 
      if(k.gt.5) go to 6 
      dep=(k-1)*25 
      if(depth.ge.dep) go to 7 
      s1=(depth+25.0-dep)*0.04 
      go to 8 
    6 dep=(k-3)*50 
      if(depth.gt.dep) go to 7 
      s1=(depth+50.0-dep)*0.02 
      go to 8 
    7 continue 
      k=17 
      s1=0.0 
    8 do 9 j=1,108 
      del=j+1 
      if(delta.ge.del) go to 9 
      s2=delta+1.0-del 
      go to 10 
    9 continue 
      j=108 
      s2=0.0 
   10 q1=q(j-1,k-1)+s1*(q(j-1,k)-q(j-1,k-1)) 
      q2=q(j,k-1)  +s1*(q(j,k)  -q(j,k-1)) 
      qval=q1+s2*(q2-q1) 
      mb=alog10(amp*.001/t)+qval  !change amp to microns for use with q tables 
c     if((t.gt.3..or.(delta.lt.15..and.depth.lt.100.)).and. 
c    1 mbflg.ne.'R') mbflg='r' 
      if((t.gt.3..or.delta.lt.15.).and.mbflg.ne.'R') mbflg='r' 
      return 
c 
   11 mbflg='n' 
      return 
      end 
= 
 
The following Q-table is used at the USGS/NEIC to compute the short-period body-wave 
magnitude mb that is published in the PDE bulletin of the USGS/NEIC.  For distances from 
5° to 109°, it represents a "digitized" version of Figure 5 of B. Gutenberg and C.F. Richter, 
1956,  "Magnitude and Energy of Earthquakes", Annali di Geofisica, v. IX, n. 1, p. 1-15.  The 
original figure is reproduced as Figure VIII-6 in Richter, C.F., 1958, Elementary Seismology, 
Freeman, San Francisco, 768 p. The origin of the Q values at distance less than 5° is not 
documented.  The USGS/NEIC does not compute mb for all distance ranges represented in the 
table. 

